

学 号:	
------	--

华中科技大学

课 程 设 计

题 目	打印水仙花数
学 院	网络空间安全学院
专 业	信息安全
班 级	
姓 名	
指导教师	

2019 年 11 月 16 日

课程设计任务书

学生姓名：_____ 专业班级：_____

指导教师：_____ 工作单位：_____

题目：打印水仙花数

初始条件：

理论：完成了《汇编语言程序设计》课程，对 80X86 系列编程结构和指令系统有了较深入的理解，已掌握了汇编语言程序设计的基本方法和技巧。

实践：完成了《汇编语言程序设计》的基本实验，熟悉了汇编语言程序的设计环境并掌握了汇编语言程序的调试方法。

要求完成的主要任务：

进一步理解和掌握较复杂程序的设计方法，掌握子程序结构、宏指令的设计和友好用户界面的设计。具体的设计任务及要求：

- 1) 键盘输入一个大整数 n ;
- 2) 将小于等于 n 的水仙花数打印出来（各位数字的立方和等于该数本身）;
- 3) 程序采用子程序结构或宏指令，结构清晰;
- 4) 友好清晰的用户界面，能识别输入错误并控制错误的修改。

在完成设计任务后，按要求撰写课程设计说明书；对课程设计说明书的具体要求请见课程设计指导书。

阅读资料：

- 1) 《80X86 汇编语言程序设计（王元珍等）》例 5.12
- 2) 《汇编语言（王爽第 3 版）》

时间安排：

1. 11 月 25 日完成程序主体
2. 11 月 26 日调试程序并完善，
3. 11 月 26 日完成该课程设计实验报告

指导教师签名：

2019 年 月 日

打印水仙花数

1. 需求说明

1.1 设计要求：

进一步理解和掌握较复杂程序的设计方法，掌握子程序结构的设计和友好用户界面的设计。

具体的设计任务及要求：

1. 输入一个三位整数 n ；
2. 将小于等于 n 的水仙花数打印出来（各位数字的立方和等于该数本身）；
3. 程序采用子程序结构或宏指令，结构清晰；
4. 友好清晰的用户界面，能识别输入错误并控制错误的修改。

1.2 设计说明

所谓“水仙花数”是指一个 3 位数，其个、十、百位数字的立方和等于其本身。

1.3 功能简述

程序可以输入一个三位数，若输入有误则提示错误，并提示重新输入，若输入正确，系统将显示出所有大于等于 100，并且小于等于输入数字的水仙花数。

然后提示是否继续，若用户输入 y 或 Y ，则系统回到最开始的状态，若输入 n 或 N ，则退出系统，若为其他则提示错误，并重新输入。

2. 设计说明（简要的分析与概要设计）

2.1 简要分析

2.1.1 原理说明

用简明扼要的文字描述说明程序包含哪些功能模块，基本的实现算法

1. 模块：

（1）输入数字处理模块

① 输入数字字符串

② 将数字字符串转换为对应数字

③ 判断数字是否为三位数

(2) 寻找水仙花数模块

① 从 100 到输入数字遍历寻找水仙花数

② 判断一个数字是否是水仙花数

③ 将数字转化为数字字符串显示

(3) 用户提示模块

① 数字输入错误重新输入

② 询问是否继续输入或是退出

2. 基本实现算法:

(1) 输入数字字符串


(2) 将数字字符串转化为数字

(3) 判断数字是否为三位数: 若不是, 输出错误提示跳至 (1)

(4) 从 100 到输入数字寻找水仙花数并将其输出, 若没有水仙花数则输出
“None”。

(5) 询问用户是否继续 (y/n): 用户输入 ‘y’ 或 ‘Y’ 则跳至 (1), 输入
‘n’ 或 ‘N’ 则程序结束, 输入其他字符输出错误提示跳至 (5)

2.1.2 程序流程图



2.2 数据段设计

data segment

;提示字符串

inputsign db 0dh,0ah,'Please input a number(100<=100<=999):\$'

inputerror db 0dh,0ah,'Input error,please input again!\$'

inputcontinue db 0dh,0ah,'Do you want to continue(y/n)? \$'

shuixianhua db 0dh,0ah,'shuixianhuashu:\$'

lastsign db 0dh,0ah,0dh,0ah,' Press any key to continue',0dh,0ah,'\$'

nohua db 'None\$'

inputstr db 5 ;存放输入数字（字符串格式）

db ?

db 5 dup(?)

mulbit db 100,10,1 ;用于计算立方和的乘数

num dw ? ;存放转换后的数字

shuixianhuashu db 10 dup(?) ;存放水仙花数（字符串格式）

data ends

3. 主要子程序（或宏指令）设计与描述

3.1 输出字符串宏定义

output macro strname

;;功能：输出字符串宏定义

;;参数：strname：输出字符串偏移地址

lea dx,strname

mov ah,9

int 21h

endm

3.2 条件转移宏定义

jcc macro cc,a,b,dest

;;功能：条件跳转宏定义

;;参数:cc:jcc 指令的具体后缀，a、b:比较的两个参数，dest:跳转到的标号

cmp a,b

j&cc dest

endm

3.3 数字字符串转化判断子程序

strtonum proc

;功能：将数字字符串转换为数字,同时判断是否介于 100 和 999 之间

;参数：si 源字符串偏移地址，di 为目的数储存位置,bx 为乘数因子位置，cx

;为数字位数

;返回值：bx 为 1 表示成功，为 0 表示失败，[di]所存为最终数字

;dx 做最终数字，al 做乘法乘数，cl 循环记数

push ax

push cx

push dx

push si

mov dx,0

mov byte ptr [di],0

bittonum:

mov al,[si] ;将每个字符移至 al

sub al,30h ;将字符转换成数字

jcc l,al,0,numerror

jcc g,al,9,numerror ;字符不在 0~9 之间则返回 0 失败

mul byte ptr [bx] ;al*[bx]即数字变为真实数值

add dx,ax ;加到最后的数字上

inc bx

inc si

loop bittonum

jcc l,dx,100,numerror ;比较最终数字小于 100 则返回 0 失败

mov [di],dx ;将最终数字移至[di]

mov bx,1 ;返回 1 成功

jmp exit

numerror:

mov bx,0

exit:

pop si

```

    pop dx
    pop cx
    pop ax
    ret
strtonum endp

```

3.4 数字转字符串显示子程序

dtoc proc

;功能：将 word 型数据转变为表示十进制数的字符串，字符串以 \$为结尾
;符，并显示

;参数：(ax)=word 型数据，ds:si 指向字符串首地址

;返回：无

```

    push ax
    push bx
    push cx
    push dx
    push si                ;主程序寄存器数据入栈保护

    mov bx,10              ;10 做除数存到 bx 中
    mov cx,0              ;cx 做计数器记位 置零
getc:
    mov dx,0              ;将 dx 清零 为除法做准备
    div bx                ;(dx,ax)除以 bx
    add dx,30h            ;余数 dx 为对应十进制数，+30h 转换为 ASCII
码
    push dx               ;将余数入栈存储
    inc cx                ;位数+1

    jcc z,ax,0,putc       ;若商为 0 跳转到 putc

    jmp getc              ;商不为 0 跳转到 getc
putc:
    pop [si]              ;将栈中字符出栈存入[si]指向的内存
    inc si                ;si 指向下一字符
    loop putc

```



```

mov byte ptr [si], ' '
mov byte ptr [si+1], '$'    ;添加结束符

pop si

mov dx, si                  ;输出数字对应的字符
mov ah, 9
int 21h

pop dx
pop cx
pop bx
pop ax                      ;主程序寄存器数据出栈恢复
ret
dtoc endp

```

3.5 寻找并输出水仙花数子程序

```

findnum proc
;功能：寻找水仙花数，并输出
;参数：
;ax 做增量, cx 记录个数, bx 做除数, dx 用来记录和

push ax
push bx
push cx
push dx
push si

mov ax, 100                ;从 100 开始遍历
mov bl, 10                 ; (bl) = 10 做除数
mov cx, 0                  ;cx 来记录水仙花数的个数
loops:                     ;总循环
push ax                    ;将当前判断数入栈保护
push cx                    ;将 cx 记录的个数入栈保护
getbit:                    ;获取每一位的数字并入栈
div bl                     ;当前判断数除以 10, ah 为余数, al 为商

```

```

mov cl,ah          ;余数入栈保护
mov ch,0
push cx
mov ah,0           ;ah 补 0
jcc nz,ax,0,getbit ;比较商是否为 0 不为 0 继续 getbit
                    ;商为 0 后

mov cx,3           ;将 cx 赋值为 3
mov dx,0           ;将和 dx 初始化为 0
addnum:            ;每一位立方和求和
    pop ax          ;余数出栈到 ax, 实际 ah 总为 0
    mov bh,al       ;此处须将 al 先复制到 bh 中储存
    mul bh
    mul bh          ;ax=bh*bh*bh
    add dx,ax       ;将每一位的立方 ax 加到 dx 中
    loop addnum

    pop cx          ;将 cx 记录的个数出栈恢复
    pop ax          ;将 ax 记录的当前判断数出栈恢复
    jcc nz,ax,dx,last ;比较 ax 和 dx 若不相等跳至 last
    lea si,shuixianhuashu
    call dtoc       ;若相等, 输出水仙花数
    inc cx          ;水仙花数个数+1
last:              ;循环变量增加
    inc ax          ;当前判断数+1
    jcc le,ax,num,loops ;比较当前判断数和 num 若 ax<=num, 继续循
环

    jcc nz,cx,0,return ;比较循环之后 cx 中记录的水仙花数
                    ;若水仙花数不为 0, 跳转到 return

    output nohua    ;若水仙花数为 0, 则输出没有

return:
    pop si
    pop dx
    pop cx
    
```

```

        pop bx
        pop ax
        ret
findnum endp

```

4. 源程序与执行结果

4.1 源程序代码

```

data segment
;提示字符串
inputsign db 0dh,0ah,'Please input a number(100<=100<=999):$'
inputerror db 0dh,0ah,'Input error,please input again!$'
inputcontinue db 0dh,0ah,'Do you want to continue(y/n)?$'
shuixianhua db 0dh,0ah,'shuixianhuashu:$'
lastsign db 0dh,0ah,0dh,0ah,'          Press any key to continue',0dh,0ah,$'
nohua db 'None$'

inputstr db 5          ;存放输入数字（字符串格式）
        db ?
        db 5 dup(?)

mulbit db 100,10,1      ;用于计算立方和的乘数
num dw ?                ;存放转换后的数字
shuixianhuashu db 10 dup(?) ;存放水仙花数（字符串格式）
data ends

stack segment
        dw 200 dup(?)
stack ends

code segment
        assume ds:data,ss:stack,cs:code

output macro strname
;;功能：输出字符串宏定义
;;参数：strname：输出字符串偏移地址

```

```

    lea dx,strname
    mov ah,9
    int 21h
endm

```

jcc macro cc,a,b,dest

;;功能：条件跳转宏定义

;;参数：cc: jcc 指令的具体后缀，a、b：比较的两个参数，dest：跳转到的标号

```

    cmp a,b
    j&cc dest
endm

```

start:

```

    mov ax,data
    mov     ds,ax
    mov ax,stack
    mov     ss,ax

```

input:

output inputsign ;输出输入数字提示

lea dx,inputstr ;输入数字（字符串格式）

mov ah,10

int 21h

jcc nz,inputstr+1,3,inputagain

;先比较输入的位数是否满足

lea si,inputstr+2

lea di,num

lea bx,mulbit

mov cl,inputstr+1

call strtonum ;将输入数字由字符串转换为数字

jcc z,bx,0,inputagain ;若 strtonum 返回值为 0 失败，则重新输入

```
output shuixianhua      ;输出水仙花标题
call findnum           ;寻找水仙花数，并输出
```

continue:

```
output inputcontinue    ;输入是否继续提示？
```

```
mov ah,1
int 21h                ;输入字符
```

```
jcc z,al,'y',input
jcc z,al,'Y',input      ;则跳转到 input 重新开始
```

```
jcc z,al,'n',ending
jcc z,al,'N',ending     ;则跳转到 ending 程序结束
```

```
output inputerror      ;若输入其他字符，则提示输入错误，重新输入
```

```
jmp continue          ;跳转到 continue 重新判断是否继续
```

ending:

```
output lastsign        ;输出最后的 press any key....
```

```
mov ah,1
int 21h                ;接受任意字符后结束程序
```

```
mov ax,4c00h
int 21h
```

inputagain:

```
output inputerror      ;输入错误，重新输入
```

```
jmp input
```

strtonum proc

;功能：将数字字符串转换为数字,同时判断是否介于 100 和 999 之间

;参数: si 源字符串偏移地址, di 为目的数储存位置,bx 为乘数因子位置, cx
;为数字位数

;返回值: bx 为 1 表示成功, 为 0 表示失败, [di]所存为最终数字

;dx 做最终数字, al 做乘法乘数, cl 循环记数

push ax

push cx

push dx

push si

mov dx,0

mov byte ptr [di],0

bittonum:

mov al,[si] ;将每个字符移至 al

sub al,30h ;将字符转换成数字

jcc l,al,0,numerror

jcc g,al,9,numerror ;字符不在 0~9 之间则返回 0 失败

mul byte ptr [bx] ;al*[bx]即数字变为真实数值

add dx,ax ;加到最后的数字上

inc bx

inc si

loop bittonum

jcc l,dx,100,numerror ;比较最终数字小于 100 则返回 0 失败

mov [di],dx ;将最终数字移至[di]

mov bx,1 ;返回 1 成功

jmp exit

numerror:

mov bx,0

exit:

pop si

pop dx

pop cx

pop ax

ret

strtonum endp

dtoc proc

;功能：将 word 型数据转变为表示十进制数的字符串，字符串以 \$ 为结尾
;符，并显示

;参数：(ax)=word 型数据，ds:si 指向字符串首地址

;返回：无

push ax

push bx

push cx

push dx

push si ;主程序寄存器数据入栈保护

mov bx,10 ;10 做除数存到 bx 中

mov cx,0 ;cx 做计数器记位 置零

getc:

mov dx,0 ;将 dx 清零 为除法做准备

div bx ;(dx,ax)除以 bx

add dx,30h ;余数 dx 为对应十进制数，+30h 转换为 ASCII

码

push dx ;将余数入栈存储

inc cx ;位数+1

jcc z,ax,0,putc ;若商为 0 跳转到 putc

jmp getc ;商不为 0 跳转到 getc

putc:

pop [si] ;将栈中字符出栈存入[si]指向的内存

inc si ;si 指向下一字符

loop putc

mov byte ptr [si],'

mov byte ptr [si+1],'\$' ;添加结束符

pop si

```

    mov dx,si                ;输出数字对应的字符
    mov ah,9
    int 21h

    pop dx
    pop cx
    pop bx
    pop ax                  ;主程序寄存器数据出栈恢复
    ret

dtoc endp

findnum proc
;功能：寻找水仙花数，并输出
;参数：
;ax 做增量， cx 记录个数,bx 做除数,dx 用来记录和
    push ax
    push bx
    push cx
    push dx
    push si

    mov ax,100              ;从 100 开始遍历
    mov bl,10               ; (bl) =10 做除数
    mov cx,0                ;cx 来记录水仙花数的个数
loops:                      ;总循环
    push ax                 ;将当前判断数入栈保护
    push cx                 ;将 cx 记录的个数入栈保护
getbit:                     ;获取每一位的数字并入栈
    div bl                  ;当前判断数除以 10， ah 为余数， al 为商
    mov cl,ah               ;余数入栈保护
    mov ch,0
    push cx
    mov ah,0                ;ah 补 0
    jcc nz,ax,0,getbit      ;比较商是否为 0 不为 0 继续 getbit

```



```

                                ;商为 0 后
                                ;将 cx 赋值为 3
    mov cx,3
                                ;将和 dx 初始化为 0
    mov dx,0
addnum:                        ;每一位立方和求和
    pop ax                      ;余数出栈到 ax, 实际 ah 总为 0
    mov bh,al                   ;此处须将 al 先复制到 bh 中储存
    mul bh
    mul bh                      ;ax=bh*bh*bh
    add dx,ax                   ;将每一位的立方 ax 加到 dx 中
    loop addnum

    pop cx                      ;将 cx 记录的个数出栈恢复
    pop ax                      ;将 ax 记录的当前判断数出栈恢复
    jcc nz,ax,dx,last           ;比较 ax 和 dx 若不相等跳至 last
    lea si,shuixianhuashu
    call dtoc                   ;若相等, 输出水仙花数
    inc cx                      ;水仙花数个数+1
last:                          ;循环变量增加
    inc ax                      ;当前判断数+1
    jcc le,ax,num,loops         ;比较当前判断数和 num 若 ax<=num, 继续循
环
                                ;比较循环之后 cx 中记录的水仙花数
                                ;若水仙花数不为 0, 跳转到 return

    jcc nz,cx,0,return

    output nohua                ;若水仙花数为 0, 则输出没有

return:
    pop si
    pop dx
    pop cx
    pop bx
    pop ax
    ret
findnum endp

```

code ends

end start

4.2 测试

4.2.1 测试用例


输入数字：

- ① 错误输入
- ② 错误输入
- ③ 正确输入

是否继续：

- ① 错误输入
- ② 错误输入
- ③ 正确输入
- ④ 正确输入

4.2.2 测试结果



The screenshot shows a DOSBox window with the title 'DOSBox 0.74, Cpu speed: 3000 cycles, Frameskip 0, Program: CD22'. The command prompt shows 'C:\>cd22'. The program output is as follows:

```
Please input a number(100<=100<=999):12
Input error,please input again!
Please input a number(100<=100<=999):045
Input error,please input again!
Please input a number(100<=100<=999):265
shuixianhuashu:153
Do you want to continue(y/n)? 1
Input error,please input again!
Do you want to continue(y/n)? t
Input error,please input again!
Do you want to continue(y/n)? y
Please input a number(100<=100<=999):999
shuixianhuashu:153 370 371 407
Do you want to continue(y/n)? n

Press any key to continue
```

5. 总结

本次汇编课程设计完成了一个具有较为完整的程序，并且运用了子程序、宏汇编等结构来使程序更加简洁、更加有可读性，进一步加强了个人用汇编语言编写程序的能力。也在编写程序的过程中，深刻感受到了汇编语言作为编程语言自身的语法特点，同时它作为低级语言又具有许多繁琐的规定要求，对程序设计有着更加苛刻的要求，此外调试也不如高级语言直接，出现问题也很难确定问题的具体出处，使调试也颇有难度。本次课程设计基本没有遇到问题，主要是在程序变量过多的时候会出现用错变量的情况，这需要个人在编程时更加细心，同时做好编程前的程序设计工作。

6. 参考资料

无

本科生课程设计成绩评定表

班级： 姓名： 学号：

序号	评分项目	满分	实得分
1	学习态度认真、遵守纪律	10	
2	设计分析合理性	10	
3	设计方案正确性、可行性、创造性	20	
4	设计结果正确性	40	
5	设计报告的规范性	10	
6	设计验收	10	
		总得分/等级	
评语：			

注：最终成绩以五级分制记。优（90-100 分）、良（80-89 分）、中（70-79 分）、及格（60-69 分）、60 分以下为不及格

指导教师签名：

2018 年 月 日